

*UPLOAD THE DATA

global controls_all kurdish_self satisfaction_pre_1 vote_regime vote_hdp age education female 
global outcomes_trust specific_trust executive_trust diffuse_trust 
global outcomes_ethnic salience ethnicdemands_2 ethnicdemands_1  ethnicdemands_3
global outcomes_ethnic languagedemands discrimination_index politicaldemands salience
global outcomes_other efficacy dem_satisfied_1 citizenship_p citizenship_a  ethnicity_p ethnicity_a


******FIGURE 1********
**********************
**********************

est clear  // clear the est locals
eststo: reg statekurdish tr_k_t
estimates store statekurdish_s

eststo: reg statekurdish tr_k_t $controls_all
estimates store statekurdish_f

eststo: reg stateservice tr_k_t
estimates store stateservice_s

eststo: reg stateservice tr_k_t $controls_all
estimates store stateservice_f

*FOR MAIN PAPER
coefplot   (statekurdish_s, label (No Controls) keep (tr_k_t) levels(95 90)) (statekurdish_f, label (Controls Added) keep (tr_k_t) levels(95 90)  mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f"))), bylabel (State Serves in Kurdish More) || (stateservice_s, keep (tr_k_t) levels(95 90)  mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f"))) (stateservice_f, keep (tr_k_t) levels(95 90)  mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f"))), bylabel (State Serves More)  ||, drop(_cons) xline(0)   byopts(compact cols(2)) subtitle(, color(black) bcolor(white)) ylabel(,labsize(medsmall)) 
  
  
  
******FIGURE 2********
**********************
**********************  
  

est clear  // clear the est locals
reg useful tr_k_t
estimates store useful_s
reg useful tr_k_t $controls_all
estimates store useful_f

coefplot (useful_s, label(Base Model) pstyle(p3)) (useful_f, label(Full Model) pstyle(p4)), drop(_cons age kurdish_self satisfaction_pre_1 vote_regime vote_hdp vote_opposition  education female) xline(0) levels(95 90)   byopts(compact cols(1))   xscale(range(-0.25 (0.25) 1)) xlabel(-0.25 (0.25) 1) subtitle(, color(black) bcolor(white)) mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f")) ylabel(,labsize(medsmall)) 
  
  
******FIGURE 3********
**********************
**********************  
    
foreach k of varlist  $outcomes_trust {
reg `k' tr_k_t 
estimates store `k'_s
reg `k' tr_k_t $controls_all
estimates store `k'_f
}

coefplot (diffuse_trust_s, label (No Controls) keep (tr_k_t) levels(95 90) mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f"))) (diffuse_trust_f, label (Controls Added) keep (tr_k_t) levels(95 90) mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f"))), bylabel (Diffuse Trust) || (executive_trust_s, keep (tr_k_t) levels(95 90) mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f"))) (executive_trust_f, keep (tr_k_t) levels(95 90)), bylabel (Executive Trust)   || (specific_trust_s, label (Simple Model) keep (tr_k_t) levels(95 90)) (specific_trust_f, label (Full Model) keep (tr_k_t) levels(95 90)), bylabel (Specific Trust) ||, drop(_cons) xline(0)   byopts(compact cols(3)) subtitle(, color(black) bcolor(white))
  
  
******FIGURE 4********
**********************
**********************   

foreach k of varlist  $outcomes_ethnic {
reg `k' tr_k_t 
estimates store `k'_s
reg `k' tr_k_t $controls_all
estimates store `k'_f
}

coefplot (languagedemands_s, label (No Controls) keep (tr_k_t) levels(95 90) mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f"))) (languagedemands_f, label (Controls Added) keep (tr_k_t) levels(95 90) mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f"))), bylabel (Language Demands) || (discrimination_index_s, keep (tr_k_t) levels(95 90) mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f"))) (discrimination_index_f, keep (tr_k_t) levels(95 90)), bylabel (Ethnic Grievances) ||  (politicaldemands_s, keep (tr_k_t) levels(95 90) mlabposition(1) mlabgap(*2) mlabel("{it:p} = " + string(@pval,"%9.3f"))) (politicaldemands_f, keep (tr_k_t) levels(95 90)), bylabel (Ethnoterritorial Demands) || , drop(_cons) xline(0)   byopts(compact cols(1)) subtitle(, color(black) bcolor(white))


*********************************************************************************
*********************************************************************************
*********************************************************************************
***************************ONLINE APPENDIX***************************************
*********************************************************************************
*********************************************************************************
*********************************************************************************

**********TABLE 1*********
**************************

sum tr_k_t tr_t_c tr_k_c stateservice statekurdish useful_e_t benefit_e_t diffuse_trust specific_trust executive_trust languagedemands discrimination_index politicaldemands salience age female college kurdish_self satisfaction_pre_1 dem_satisfied_1 vote_regime vote_hdp vote_opposition efficacy citizenship_p citizenship_a ethnicity_p ethnicity_a

**********TABLE 2*********
**************************

tab region_recoded

**********TABLE 3*********
**************************

est clear  // clear the est locals
eststo: reg tr_k_t $controls_all
eststo: reg tr_k_t $controls_all if Progress>90

esttab using "regression.tex", b(3) se(3) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) label booktabs collabels(none)  mtitles("Entire Sample" "Only Completes") nolines noconstant nonum alignment(D{.}{.}{-1})   replace


**********TABLE 4*********
**************************

 est clear  // clear the est locals
eststo: reg statekurdish tr_k_t
eststo: reg statekurdish tr_k_t $controls_all
eststo: reg statekurdish tr_k_c
eststo: reg statekurdish tr_k_c $controls_all
eststo: reg statekurdish tr_t_c
eststo: reg statekurdish tr_t_c $controls_all

eststo: reg stateservice tr_k_t
eststo: reg stateservice tr_k_t $controls_all
eststo: reg stateservice tr_k_c
eststo: reg stateservice tr_k_c $controls_all
eststo: reg stateservice tr_t_c
eststo: reg stateservice tr_t_c $controls_all


esttab using "regression.tex", b(3) se(3) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) label booktabs collabels(none) mgroups("State Kurdish" "State Services", pattern(1 0 0 0 0 0 1 0 0 0 0 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))  mtitles("Solo" "Full" "Solo" "Full") nolines noconstant alignment(D{.}{.}{-1})   replace


**********TABLE 5*********
**************************

est clear
eststo: reg useful tr_k_t
eststo: reg useful tr_k_t $controls_all

esttab using "regression.tex", b(3) se(3) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) label booktabs collabels(none)  m nolines noconstant nonum alignment(D{.}{.}{-1})   replace

**********TABLE 6*********
**************************

est clear
foreach k of varlist  $outcomes_trust {
eststo: reg `k' tr_k_t 
eststo: reg `k' tr_k_t $controls_all
}

esttab using "regression.tex", b(3) se(3) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) label booktabs collabels(none) mgroups("Specific Trust" "Executive Trust" "Diffuse Trust", pattern(1 0 1 0 1 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))  nomtitles nolines noconstant alignment(D{.}{.}{-1})   replace


**********TABLE 7*********
**************************

est clear
foreach k of varlist  $outcomes_ethnic {
eststo: reg `k' tr_k_t 
eststo: reg `k' tr_k_t $controls_all
}

esttab using "regression.tex", b(3) se(3) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) label booktabs collabels(none)   nomtitles nolines noconstant alignment(D{.}{.}{-1}) mgroups("Language Demands" "Grievances" "Ethnoterritorial Demands" "Issue Salience", pattern(1 0 1 0 1 0 1 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))   replace


**********TABLE 8*********
**************************

est clear
foreach k of varlist  $outcomes_other {
eststo: reg `k' tr_k_t 
eststo: reg `k' tr_k_t $controls_all
}

esttab using "regression.tex", b(3) se(3) star(+ 0.10 * 0.05 ** 0.01 *** 0.001) label booktabs collabels(none)   nomtitles nolines noconstant alignment(D{.}{.}{-1}) mgroups("efficacy" "dem_satisfied_1" "citizenship_p" "citizenship_a"  "ethnicity_p" "ethnicity_a", pattern(1 0 1 0 1 0 1 0 1 0 1 0) prefix(\multicolumn{@span}{c}{) suffix(}) span erepeat(\cmidrule(lr){@span}))   replace















